# エクゾチックオプション価格のシミュレーション　その３ －スプレッドオプションの価格式－

**Numerical Tests on the Exotic Option Pricing Models － 3 －**
**The Spread Option Pricing Model**

慶応義塾大学理工学部
竹内寿一郎

## １．はじめに

昨年のシンポジウムでブラック・ショールズの式に基づくエクゾチックオプションの中で、最も基本となるデジタルオプションをとりあげたが【1】、ここでは２つの資産をとりあげ、その資産の価格の差にたいするオプション価格について論ずることにする。【3】

これまでの資産のオプション価格と大きく異なる点は、二つの資産の差であるから、負の値をも許すということである。

## ２．スプレッドオプション価格式の仮定

二つの資産 $S_1$、$S_2$ 現物の価格はそれぞれ幾何ブラウン運動するとして、その価格の差 $S(t)$ は

(1)　$S(t) = S_1(t)e^{(r-q_1)(T-t)} - S_2(t)e^{(r-q_2)(T-t)}$

と表せる。ここでこの $S$ が中立世界において次のような正規過程に従うものと仮定する。

(2)　$dS = \sigma d\hat{z}$

このときこのスプレッドコールオプションの評価式を求めるのには $Max\{S(T)-K,0\}$ の期待値、スプレッドプットオプションの評価式を求めるのには $Max(K-S(T),0)$ の期待値を計算すればよいことになる。

(2) 式から $S(t)$ の変化は平均 0、分散 $\sigma^2 dt$ の正規過程に従うので、$t$ からオプション満期である $T$ 時点までの $S(T)$ の変化は次のように書くことができる。

(3)　$$\frac{S(T)-S(t)}{\sigma\sqrt{T-t}} \quad \sim \quad \omega$$

ここで、$\omega$ は標準正規分布に従う確率変数である。

## ３．コールオプションの期待値

オプション満期 $T$ におけるスプレッド価格 $S(T)$ の確率密度関数を $f(S(T))$ とすると、ペイオフ $Max\{S(T)-K,0\}$ の期待値 $C^*$ は、

(4)　$$\begin{aligned} C^* &= \int_{-\infty}^{\infty} Max\{S(T)-K,0\}f(S(T))dS(T) \\ &= \int_{K}^{\infty} \{S(T)-K\}f(S(T))dS(T) \end{aligned}$$

(4) 式の 2 行目は $Max$ をはずす為に、積分範囲を ($-\infty$ から $\infty$) を ($K$ から $\infty$) に変えたものである。確率変数 $S(T)$ を (3) 式により規準化し、$\omega$ を $x$ とおき直して (4) 式を書き直すと、

(5)　$$C^* = \int_{x_0}^{\infty} \{S(t) + x\sigma\sqrt{T-t} - K\}\phi(x)dx$$

ここで、(3) より

$$(6)\quad \begin{cases} S(T) & = & S(t) + x\sigma\sqrt{T-t} \\ x_0 & = & \dfrac{K - S(t)}{\sigma\sqrt{T-t}} \end{cases}$$

を用いた。
(5) 式より、

$$(7)\quad C^* = \sigma\sqrt{T-t}\int_{x_0}^{\infty} x\phi(x)dx + \{S(t) - K\}\int_{x_0}^{\infty}\phi(x)dx$$

標準正規分布の積分に関する表は用意されているので、第 2 項は計算できるが、第 1 項は $x$ と密度関数の積の積分を求めねばならない。しかしこの第 1 項は $\phi(x)$ の微分 $\phi(x)' = -x\phi(x)$ を利用すると、

$$(8)\quad \int x\phi(x)dx = -\phi(x)$$ なる関係を使って

結局

$$(9)\quad \begin{cases} C^* & = & \sigma\sqrt{T-t}\,[-\phi(x)]_{x_0}^{\infty} + \{S(t) - K\}\int_{x_0}^{\infty}\phi(x)dx \\ & = & \sigma\sqrt{T-t}\phi(x_0) + \{S(t) - K\}\int_{-\infty}^{-x_0}\phi(x)dx \\ & = & \sigma\sqrt{T-t}\phi(-x_0) + \{S(t) - K\}\Phi(-x_0) \end{cases}$$

満期 $T$ 時点の価格 $C*$ を現在のオプション価格 $C$ に直して、

$$(10)\quad C = e^{-r(T-t)}C^* = e^{-r(T-t)}\left[\sigma\sqrt{T-t}\phi(-x_0) + \{S(t) - K\}\Phi(-x_0)\right]$$

ここに改めてスプレッドコールオプションの現在価格は、リスクフリーな金利を $r$ として、

$$(11)\quad C = e^{-r(T-t)}\left[\sigma\sqrt{T-t}\phi\left(\frac{S(t) - K}{\sigma\sqrt{T-t}}\right) + \{S(t) - K\}\Phi\left(\frac{S(t) - K}{\sigma\sqrt{T-t}}\right)\right]$$

と書ける。ここで $\phi(x)$ は標準正規分布の密度関数、$\Phi(x)$ は標準正規分布の累積分布関数である。

## 4．プットオプションの期待値

プットオプションはペイオフ $Max(K - S(T), 0)$ の期待値であるから、

$$(12)\quad P^* = \int_{-\infty}^{\infty} Max\{K - S(t), 0\}f(S(T))dS(T)$$
$$= \int_{-\infty}^{K}\{K - S(T)\}f(S(T))dS(T)$$

ここで (3) 式の変換を行い、$\omega$ を $x$ とおき、(6) を使うと、

$$(13)\quad P^* = \int_{-\infty}^{x_0}\{K - S(t) - x\sigma\sqrt{T-t}\}\phi(x)dx$$
$$= \{K - S(t)\}\int_{-\infty}^{x_0}\phi(x)dx - \sigma\sqrt{T-t}\int_{-\infty}^{x_0}x\phi(x)dx$$
$$= \{K - S(t)\}\Phi(x_0) - \sigma\sqrt{T-t}\,[-\phi(x)]_{-\infty}^{x_0}$$
$$= \{K - S(t)\}\Phi(x_0) + \sigma\sqrt{T-t}\phi(x_0)$$
$$= \sigma\sqrt{T-t}\phi(x_0) - \{S(t) - K\}\Phi(x_0)$$

改めてスプレッドプットオプションの現在価格 $P$ は、

$$(14)\quad P = e^{-r(T-t)}\left[\sigma\sqrt{T-t}\phi\left(\frac{K - S(t)}{\sigma\sqrt{T-t}}\right) - \{S(t) - K\}\Phi\left(\frac{K - S(t)}{\sigma\sqrt{T-t}}\right)\right]$$

と書ける。

## 5．スプレッドコールオプション価格の J プログラム

【実行例】100 円の差に対し、12ヵ月後の 110 円の差でコールオプション価格について、1000 回のシュミレーションを 5 回繰り返してみた。ただし、期間は 12ヶ月、ボラリティは 20 円、
リスクフリー金利は 5 パーセントとした。

```
   1000 Spread 100 110 12 20 5
3.71013
   1000 Spread 100 110 12 20 5
3.36959
   1000 Spread 100 110 12 20 5
3.52783
   1000 Spread 100 110 12 20 5
3.33695
   1000 Spread 100 110 12 20 5
3.74484
```

計算式 (11) による結果は、

```
   Sp 100 110 12 20 5
3.763
```

であった。

【Ｊによる関数リスト】

```
NB.*****************************************************
NB. Spread Option Pricing Model-Simulation
NB. NN Spread Sdif Kdif Term Sig R
NB. Usage:1000 Spread 100 150 12 20 5
NB.*****************************************************
   Spread=:4 : 0
'Sdif Kdif T Sig r'=.y.
i=.0[S=:MM#Sdif[MM=.x.
label_L1.
if. T<i=.>:i do. goto_owari. end.
z=.Rndm_Norm MM,0,1
S=.S+(Sig%%:12)*z
NB. print S
goto_L1.
label_owari.
w=.S-Kdif
C=.(^-(r%100)*(T%12))*MM%~+/(0<w)*w
)
```

```
NB.**************************************************
NB. Normal Random Numbers  Usage:Rndm_Norm 10000 5 2
NB.    Rndm_Norm Size Mu Sigma
NB.**************************************************
   Rndm_Norm=:3 : 0
'Num Mean Sigma'=.y.
Mean+Sigma*Ninv_y"0 (?Num#10000000)%10000000
)
NB. Yamanouti's Formula
   Ninv_y=:3 : 0
z=.-^.4*y.*(1-y.)
x=.%:z*(2.0611786-5.7262204%(z+11.640595))
if. y.>0.5 do. x=.-x end.
)
NB. **************************************************
NB. Normal distribution
NB. J.Takeuchi 1972 Statistical Tables
NB. Japanese Industrial Standard Association
NB. *************************************************
stnormal=:(%:@o.@2:)(%~)^@-@-:@*:
   NP=:3 : 0
(stnormal y.)*y.%(-`%`+`%)/,(>:+:k),.(*:    y.)*>:k=.i.28)
)
   NQ=:3 : 0
(stnormal y.)*%`+/1,,y. ,.>:i.28
)
   Ndist=:3 : 0
if. 3.3<z=:|y. do. q=:NQ z else. q=:0.5-NP z end.
if. 0<:y. do. q=:1-q end.
)
NB. ==========================================
NB. Spread Option Pricing Model
NB. J.Takeuchi Apr. 2007
NB. Usage: Sp data
NB. data is list ( 5 menbers)
NB. PriceDif ExPriceDif  Term(Month) Sigma FreeRate
NB. ex. data=. 100 120 12 20 6
NB. ==========================================
Sp =: 3 : 0
'Pdif Kdif Term Sig Rate' =: y.
```

```
rt=.%:t=. Term % 12
u=. (Pdif-Kdif) % Sig*rt
p1=. Sig*rt*stnormal u
p2=. (Pdif-Kdif)*Ndist u
Sp=. (p1+p2)*( ^ (-Rate%100) * t)
)
```

【参考文献】

【1】竹内寿一郎・本田皓士（2006)：エキゾチックオプション価格のシミュレーション　その１　ーキャッシュデジタルとプレインバニラオプションー　JAPLA 2006 シンポジウム　2006.12.9 資料

【2】竹内寿一郎（2007)：エキゾチックオプション価格のシミュレーション　その２　ーダイアログボックスで決めるオプション価格のＪ関数ー　JAPLA 研究会 2007.1.27 資料

【3】山下司 (2001)：オプションプライシングの数理　ー基礎理論と専門書のブリッジテキストー　金融財政事情研究会